Supplementary problems for #6.30

1. In one particular regression model building procedure, the jth variable is removed from the regression model if a t-test for (j does not reject Ho.  For model #2, notice that facilities is not significantly related to stay.  Remove this variable from the model and find the new model.  

Part of the SAS output needed to answer the above question:

                           Parameter Estimates

                      Parameter       Standard

 Variable     DF       Estimate          Error    t Value    Pr > |t|

 Intercept     1        6.27035        0.50388      12.44      <.0001

 beds          1        0.00247     0.00082361       3.00      0.0033

 infection     1        0.63238        0.11845       5.34      <.0001
2. Calculate R2 and 

.  What happens to these measures?

Part of the SAS output needed to answer the above question:

         Root MSE              1.56832    R-Square     0.3388

         Dependent Mean        9.64832    Adj R-Sq     0.3268

     Coeff Var            16.25486
3. Create the same plots as done for #6.30e.  Note that you will no longer have a plot for ei vs. facilities.  What happens to these plots for the new model?  

Partial answer: They should appear approximately the same.  

4. Using PROC G3D, create a scatter plot for the variables beds, infection, and stay.   

5. Using PROC G3D, create a plot of the estimated regression plane using beds, infection, and stay.

6. Using PROC GCONTOUR, create a contour plot of the estimated regression plane using beds, infection, and stay.

7. Create the correct plot to show that observation #47 and #112 are outliers.  Remove these observations from the data set and repeat the analyses for 1.-6.  What differences do you notice?  

Partial answer: Below are observations #47 and #112 

                  Obs    beds    infection     stay

                   47     306       6.5       19.56

                  112     835       5.9       17.94

The estimated regression plane was affected by these outliers.  You need to figure out how!  

Below is the part of the SAS program needed for these supplementary problems.  

data set3;

  infile 'c:/chris/osu/stat4043/NKNW_data/apc1.dat';  

  input ID stay age infection culture xray beds school region census nurses facilities; 

  obs_numb  = _n_;

  *RUN THIS when you want to remove 47 and 112;
  *if obs_numb = 47 or obs_numb=112 then delete;
run;

title2 'Print observations 47 and 112';

proc print data=set3;

  where obs_numb = 47 or obs_numb=112;

  var beds infection stay;

run;

title2 '#6.30: New estimated regression model';

proc reg data=set3;

  model stay = beds infection;

  output out=out_set3 r=residual p=Y_hat;

  symbol1 v=dot h=.1 cv=blue; 

  plot residual.*beds / nostat vref=0 cvref=red;; *Check linearity of model;
  plot residual.*infection / nostat vref=0 cvref=red;; *Check linearity of model;
  plot residual.*predicted. / nostat vref=0 cvref=red;  *Check error variance;
run;

*Create semi-studentized residuals;
data out_set4;

  set out_set3;

  semi_stud_resid = residual/1.56832;  *sqrt(MSE)=1.56832 when obs. #47 and #112 are included;
run;

title2 '#6.30: Examine the semistudentized residuals';

proc univariate data=out_set4 plot;

  var semi_stud_resid;

run;

*Create values for the estimated regression plane.;
data reg_plane;

  do beds = 0 to 850 by 50; *Note: 29<=beds<=833;
    do infection = 0 to 8 by 1; *Note: 1.3<=infection<=7.8;
      *stay_hat =  6.78529 + 0.00169*beds + 0.52789*infection;  *without 47 and 112;
      stay_hat =  6.27035  + 0.00247*beds + 0.63238*infection; *with 47 and 112;
      output;

    end;

  end;

run;

*Find min and max of stay;
title2 'Min and max of stay';

proc means data=set3;

  var stay;

run;

*Create plot of the estimated regression plane;
proc g3d data=reg_plane;

  title2 '3D scatter plot';

  plot beds*infection=stay_hat / grid zmin=5 zmax=20 xticknum=9 yticknum=18 zticknum=16;

run;

*Create scatter plot;
proc g3d data=set3;

  title2 '3D scatter plot';

  scatter beds*infection=stay / grid shape='cube' zmin=5 zmax=20 zticknum=16 color='blue';

run;

*Create contour plot;
proc gcontour data=reg_plane;

  plot beds*infection=stay_hat / levels=6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 

                                 autolabel=(reveal) 

                               haxis=axis1 vaxis=axis2 legend=legend1 grid ;

  title2 'Contour plot';                                

  axis1 label = ('Infection')

        length=10 

        order = (0 to 8 by 1);

  axis2 label=(a=90 'Beds')

        length=10 

        order = (0 to 850 by 50);

  legend1 label=('Predicted Stay')

          down=2;

run;

